

## HT48 MCU 的 WDT 的使用

文件编码：HA0019s

### 简介

在 HT48 系列单片机内部有一个看门狗电路，主要是用来防止应用程序跳到未知的地方或是进入死循环。本文以 HT48R10A-1 为例，介绍 HT48RXA-1 系列中 WDT 的使用方法及其注意事项。

### 使用说明

- (1) WDT 的前置分频寄存器是 WDTS(09H)，WDT 的溢出会置位 TO 标志；执行“CLR WDT”会清除 TO 标志，“CLR WDT”指令同时会清除 PD 标志。WDT 可由掩膜选项禁用，禁用后，所有对 WDT 的操作将无效。
- (2) 看门狗定时器的时钟来源由掩膜选项来决定，可以选择为 WDT 时钟 (WDTCLK)、实时时钟 (RTC) 或是指令时钟 (系统时钟/4)。WDT 的计数溢出将使芯片“复位”，各值可见芯片说明中的表，但在 HALT 模式下，将使芯片“热复位”(只有 PC 和 SP 会被置零)。
 

在 HT-IDE2000 中选择掩膜选项时，应先选取 OSC 项，这样，对应的时钟来源才会在 WDT 项出现，最后，要选好 WDTinstr 项以确定清除 WDT 的指令方式。

RTC 时钟仅在内部 RC 振荡器+RTC 模式中有效。如果看门狗定时器的时钟来源为指令时钟，则看门狗定时器在 HALT 模式中会停止计数并失去保护功能。在这种情况下，只能通过外部逻辑来重新启动系统 (唤醒或置位)，所以最好使用内置的 RC 振荡器(WDT 振源)或 32KHZ 的 RTC 作时钟来源。
- (3) 有三种方式可以清除看门狗定时器的内容，分别是：外部复位、软件指令和暂停指令这三种方式。其中，软件程序指令包括两种，分别为“CLR WDT”和一组“CLR WDT1”、“CLR WDT2”指令，应用时只能选择其中的一种，由掩膜选项来决定。如果选的是“CLR WDT”(即 CLR WDT 次数为一)的话，只要执行“CLR WDT”指令即会清除看门狗定时器的内容。而如果选的是“CLR WDT1”和“CLR WDT2”这组指令 (即 CLR WDT 次数为二)，这两个指令必须交换执行才能清除看门狗定时器的内容，否则，看门狗定时器会因为超时而将系统复位。
- (4) WDT 的溢出时间，由 WDT 的时钟来源和选定的 WDTS 寄存器的分频系数决定。(见表) WDTS 的高五位是为使用者预留的特殊标志。

**看门狗定时器分频系数选择**

WS2	WS1	WS0	分频率
0	0	0	1 : 1
0	0	1	1 : 2
0	1	0	1 : 4
0	1	1	1 : 8
1	0	0	1 : 16
1	0	1	1 : 32
1	1	0	1 : 64
1	1	1	1 : 128

溢出时间计算公式如下：

$$TOVER = \text{WDT 时钟源单位时间} \times \text{八阶预分频}(256) \times \text{前置分频系数(由 WDTS 决定)}$$

例如：WDT OSC 时钟源的最大溢出周期 = 130 μs (最大单位时间 = 1 / 最小时钟频率) × 256(固

定值) $\times 128$ (最大分频系数)=4.26 S

SYSTEM FREQUENCY/4 时钟源的最小溢出周期=0.5  $\mu$  s (最快指令周期) $\times 256$ (固定值) $\times 1$ (最小分频系数)=128  $\mu$  s

各种时钟源的 WDT 的最大、最小溢出时间：

时钟来源	最大溢出周期	最小溢出周期
WDTOSC ( RC )	4.26 s	8.19 ms
32KHz ( RTC )	1.00 s	7812.5 $\mu$ s
System Frequency/4	327.68 s	128 $\mu$ s

此表值为  $V_{DD} = 5 V$  时理论推导参数，其中：最大溢出周期采用相应时钟源的最大单位时间，分频选择为 128。最小溢出周期采用相应时钟源的最快单位时间，分频选择为 1。表值未考虑温度等外界环境的影响，仅供参考

## 例程

<例 1>

程序名：wdttimeover.asm

作者：黄山云

目的：介绍 Wdt 的基本使用

掩膜选项：wdtinstr 选项选 one clear instruction

程序清单：

```

INCLUDE HT48R10A-1.INC
data .section 'data'
    count1 db  ?
code .section 'code'
org 10h          ;这是第一个程序分区
sz status.5     ;TO没有溢出则进入死循环
jmp $           ; ( 1 ) 溢出则等待
jmp start2
start: nop
    clr  wdt     ;该指令将防止WDT的溢出
start2:         ;死循环区 * * * * *
    nop
    mov a,03h   ;设分频系数为128
    ;clr wdt    ;若放在此处将无法使系统复位
    mov wdts,a  ;设定WDT定时时间
    jmp start2  ;设置的死循环 * * * * *

```

程序说明：该程序表明 WDT 的最基本应用，运行该程序一段时间后，程序指针将停于 ( 1 ) 处，表明 WDT 避免了死循环的发生。同时表明当死循环中包含了仅有的一条 WDT 清除指令时，程序仍然无法正常复位。

<例 2>

;程序名：Wdtttimeover1.asm

;作者：黄山云

;目的：该程序表明了使用一组(两条)CLR WDT 指令时，程序的结构

```

; 掩膜选项：wdtinstr 选项选 two clear instruction
; 程序清单：
INCLUDE HT48R10A-1.INC
data .section 'data'
count1 db ?
code .section 'code'
org 10h ;这是第一个程序分区
sz status.5 ;TO没有溢出则进入死循环
jmp $ ;(1) 溢出则等待
jmp start2
start: nop
clr wdt2 ;(2) 该指令将防止WDT的溢出
start2: ;死循环区*****
nop
mov a,03H ;设分频系数为128
clr wdt1 ;(3) 死循环包含了一条CLR WDT指令
mov wdts,a ;设定WDT定时时间
jmp start2 ;(4) 设置的死循环*****

```

程序说明：程序在(4)处设置了一死循环，该死循环包含了(3)处的WDT计数清除指令“CLR WDT1”，但由于(2)处的“CLR WDT2”指令的不在循环内，WDT计数器仍继续计数，程序仍将被WDT复位，最终程序会停在(1)处，从而避免了死循环的发生。

注意，使用一组清除指令时，MASK OPTION/WDTINSTR 需选定清除WDT的指令个数为“TWO CLEAR INSTRUCTIONS”。否则，将按一条清除WDT执行。本程序将无法回到预定的(1)处。

WDT溢出后，程序将由000H开始执行，所以，如果在程序中使用了ORG指令进行了程序分区，一定要注意是从第一个分区开始重新执行程序。

程序也表明，当死循环中含有CLR WDT指令时，无法实现复位，必须采用两条CLR WDT指令结合的办法，才能正确复位。

### <例 3>

```

; 程序名：wdttime.asm
; 作者：黄山云
; 目的：演示溢出时间的设定，并测量
; 掩膜选项：osc 为 32khzRTC+BUILTIN RC，TMR__CLK 为 32KHZRTC
; 程序清单：
include ht48R10A-1.inc
; ;-----
data .section 'data'
count1 equ [40h]
count2 equ [41h]
temp equ [42h]
; ;-----
code .section at 0h 'code'

```

```

org    00h
    jmp start
org    08h
    jmp intstart
start:
    sz STATUS.5
    Jmp $
    clr count1
    clr count2
    clr temp
    mov a,07H           ;设分频系数为128
    mov WDTs,a         ;设定WDT定时时间
main:
    mov a,84h          ;采用fsys=32krtc
    mov tmrc,a
    set intc.0         ;开总中断
    set intc.2         ;开定时器
    mov a,00h         ;初值为0
    mov tmr,a
    set tmrc.4        ;允许计数
$2:
    snz status.5      ;等WDT溢出
    jmp $2
    clr tmrc.4        ;停止计数
    jmp $
intstart:
    inc count1         ;每256MS计数一次
    sz count1
    jmp $1
    inc count2         ;可计算更长的延时
    sz count2
    jmp $1
    jmp $
$1:
    reti
end

```

程序说明：本程序将分别计算各个不同时钟来源所决定的溢出时间：由于 WDT 溢出复位，并不影响 TMR 寄存器，所以，可以利用定时器测量 WDT 的各个模式的溢出时间。程序定时器的时钟源为 32KRTC，即每 1ms 计数一次。当 WDT 溢出时，将置位 TO 标志，程序将不断监测 TO 标志，一旦 TO 被置位将停止定时器的计数。

当时钟来源选择 WDT 时钟(正常情况下)，RC 振荡器的周期为 78 μS/5V。

理论值：

时钟来源	看门狗溢出周期	掩膜选择
WDTOSC	$78 \mu s \times 256 \times 128 = 2.5559s$	WDT 为 RC OTC
32KHz ( RTC )	$31.25 \mu s \times 256 \times 128 = 1.024s$	WDT 为 32KHZ RTC
System Frequency/4	$1 \times 256 \times 128 = 0.032768s$	WDT 为 T1

注：System Frequency 为 4000KHz  
 WDTS 的分频系数选择为 128

测量值：

其计算公式如下：

$$TOVER = 1ms \times 256 \times COUNT1 + TMR$$

时钟来源	看门狗溢出周期	相对误差
WDTCLK	$9 \times 256 + 256 = 2.56S$	0.1%
32KHz ( RTC )	$4 \times 256 + 1 = 1.025S$	0.1%
System Frequency/4	21H=33 ms	0.7%

程序检讨：本程序未考虑程序指令执行的误差，此外，时基信号的波动也会导致误差的出现。